Linear Trapezoidal State Variable Filter (SVF) in state increment form: state += val

© Andrew Simper, Cytomic, 2014, andy@cytomic.com
created: 7th June 2014
updated: 1st Feb 2016 - switched notch and peak responses in implementation check to match psuedo code)
updated: 5th Feb 2016 - cleaned up df1 matching section to more clearly show important formula
updated: 11th Feb 2016 - using peak = high - low to keep frequencies above cutoff in phase

Putting it in the form of a state increment keeps the all coefficients near zero at low frequencies, which is feels intuitively right to me since you just increment a bit of signal onto your existing state to do your integration.

I haven't had time to add too much detail here, but please refer to the other papers located at www.cytomic.com/technical-papers for derivations of trapezoidal integration and the maths of circuit solving.

Thanks to Teemu Voipio (mystran) for pointing out the numerical benefits of using sin to calcualte the coeffecients.

Solve the nodal circuit equations for an implicit linear SVF

```
Remove["Global`*"]
sln = Solve[{0 == g (v0 - k v1 - v2) - (gc (v1 - 0) - ic1eq),
        0 == g v1 - (gc (v2 - 0) - ic2eq)} /. {gc → 1}, {v1, v2}][[1]] // FullSimplify;
sln1 = a /. Solve[(v1 /. sln) == ic1eq + a, a][[1]];
sln2 = a /. Solve[(v2 /. sln) == ic2eq + a, a][[1]];
{v1 → ic1eq + sln1}
{v2 → ic2eq + sln2}
coef0 = Coefficient[v1 /. sln, {v0, ic1eq, ic2eq}] // FullSimplify;
coef1 = Coefficient[sln1, {v0, ic1eq, ic2eq}] // FullSimplify
coef2 = Coefficient[sln2, {v0, ic1eq, ic2eq}] // FullSimplify
```

$$\left\{v1 \to ic1eq + \frac{-g^2\, ic1eq - g\, ic2eq - g\, ic1eq\, k + g\, v0}{1 + g^2 + g\, k}\right\}$$

$$\left\{v2 \to ic2eq + \frac{g\, ic1eq - g^2\, ic2eq + g^2\, v0}{1 + g^2 + g\, k}\right\}$$

$$\left\{\frac{g}{1 + g\,(g + k)},\ -1 + \frac{1}{1 + g\,(g + k)},\ -\frac{g}{1 + g\,(g + k)}\right\}$$

$$\left\{\frac{g^2}{1 + g\,(g + k)},\ \frac{g}{1 + g\,(g + k)},\ -\frac{g^2}{1 + g\,(g + k)}\right\}$$

Coefficients using Tan

```
{g0 → coef1[[1]] /. {g → Tan[π w]}}
{g1 → coef1[[2]] /. {g → Tan[π w]}}
{g2 → coef2[[1]] /. {g → Tan[π w]}}
```

$$\left\{g0 \to \frac{\text{Tan}[\pi\, w]}{1 + \text{Tan}[\pi\, w]\, (k + \text{Tan}[\pi\, w])}\right\}$$

$$\left\{g1 \to -1 + \frac{1}{1 + \text{Tan}[\pi\, w]\, (k + \text{Tan}[\pi\, w])}\right\}$$

$$\left\{g2 \to \frac{\text{Tan}[\pi\, w]^2}{1 + \text{Tan}[\pi\, w]\, (k + \text{Tan}[\pi\, w])}\right\}$$

Coefficients in terms of Sin instead of Tan

```
{g0 -> FullSimplify[coef1[[1]] /. {g → s / c}] /.
    {c → Cos[π w], s → Sin[π w]} // Simplify}
ss = FullSimplify[coef1[[2]] /. {g → s / c}] /. {c → Cos[π w], s → Sin[π w]} //
    Simplify;
{g1 → - (Expand[Numerator[ss]] /. {Cos[π w] Sin[π w] → 1 / 2 Sin[2 π w]}) /
    Denominator[ss]}
{g2 -> FullSimplify[coef2[[1]] /. {g → s / c}] /.
    {c → Cos[π w], s → Sin[π w]} // Simplify}
```

$$\left\{g0 \to \frac{\text{Sin}[2\,\pi\, w]}{2 + k\, \text{Sin}[2\,\pi\, w]}\right\}$$

$$\left\{g1 \to \frac{2\, \text{Sin}[\pi\, w]^2 + k\, \text{Sin}[2\,\pi\, w]}{2 + k\, \text{Sin}[2\,\pi\, w]}\right\}$$

$$\left\{g2 \to \frac{2\, \text{Sin}[\pi\, w]^2}{2 + k\, \text{Sin}[2\,\pi\, w]}\right\}$$

Solve coefficients in terms of Sin instead of Tan with scaling of A

```
ss =
 FullSimplify[coef1[[1]] /. {g → A s / c}] /. {c → Cos[π w], s → Sin[π w]} // Simplify
{g0 -> FullSimplify[coef1[[1]] /. {g → A s / c}] /.
   {c → Cos[π w], s → Sin[π w]} // Simplify}
ss = FullSimplify[coef1[[2]] /. {g → A s / c}] /.
    {c → Cos[π w], s → Sin[π w]} // Simplify;
{g1 → - (Expand[Numerator[ss]] /. {Cos[π w] Sin[π w] → 1 / 2 Sin[2 π w]}) /
   Denominator[ss]}
{g2 -> FullSimplify[coef2[[1]] /. {g → A s / c}] /.
   {c → Cos[π w], s → Sin[π w]} // Simplify}
```

$$\frac{A\,\mathrm{Cos}[\pi w]\,\mathrm{Sin}[\pi w]}{\mathrm{Cos}[\pi w]^2 + A\,k\,\mathrm{Cos}[\pi w]\,\mathrm{Sin}[\pi w] + A^2\,\mathrm{Sin}[\pi w]^2}$$

$$\left\{g0 \to \frac{A\,\mathrm{Cos}[\pi w]\,\mathrm{Sin}[\pi w]}{\mathrm{Cos}[\pi w]^2 + A\,k\,\mathrm{Cos}[\pi w]\,\mathrm{Sin}[\pi w] + A^2\,\mathrm{Sin}[\pi w]^2}\right\}$$

$$\left\{g1 \to 1 - \frac{1}{1 + A\,k\,\mathrm{Tan}[\pi w] + A^2\,\mathrm{Tan}[\pi w]^2}\right\}$$

$$\left\{g2 \to \frac{A^2\,\mathrm{Sin}[\pi w]^2}{\mathrm{Cos}[\pi w]^2 + A\,k\,\mathrm{Cos}[\pi w]\,\mathrm{Sin}[\pi w] + A^2\,\mathrm{Sin}[\pi w]^2}\right\}$$

Plot the coefficients to check the sin version matches the tan one

Note that the coefficients are near zero when the cutoff is near zero, and they remain bounded

```
Show[Table[Plot[{Tan[π w]/(1 + Tan[π w] (Tan[π w] + k)),
    -1 + 1/(1 + Tan[π w] (Tan[π w] + k)), Tan[π w]^2/(1 + Tan[π w] (Tan[π w] + k))},
   {w, 0, 1/2}], {k, 0, 2}], PlotLabel → "Tan form"]

Show[Table[Plot[{Sin[2 π w]/(2 + k Sin[2 π w]), -(k Sin[2 π w] + 2 Sin[π w]^2)/(2 + k Sin[2 π w]), 2 Sin[π w]^2/(2 + k Sin[2 π w])},
   {w, 0, 1/2}], {k, 0, 2}], PlotLabel → "Sin form"]
```





Psuedo code

```
init sin :
k = damp
w = pi*cutoff/samplerate
s1 = sin (w)
s2 = sin (2*w)
nrm = 1/(2 + k*s2)
g0 = (s2)*nrm
g1 = (-2*s1*s1 - k*s2)*nrm
g2 = (2*s1*s1)*nrm

init tan :
k = damp
w = pi*cutoff/samplerate
g = tan (w)
g0 = g/(1 + g*(g + k))
g1 = (-g - k)*g0
g2 = g*g0

clear :
```

```
ic1eq = 0
ic2eq = 0

tick :
v0 = input
t0 = v0 - ic2eq
t1 = g0*t0 + g1*ic1eq
t2 = g2*t0 + g0*ic1eq
v1 = t1 + ic1eq
v2 = t2 + ic2eq
ic1eq = ic1eq + 2.0*t1
ic2eq = ic2eq + 2.0*t2
high = v0 - k*v1 - v2 (7* 9+ = 16 ops)
band = v1 (6* 7+ = 13 ops)
low = v2 (6* 7+ = 13 ops)
notch = high + low = v0 - k*v1 (7* 8+ = 15 ops)
peak = high - low = v0 - k*v1 - 2*v2 (8* 9+ = 17 ops)
```

Implementation check against tan version

```
CalcCoeff1[cutoff_, damp_, sr_] := Block[{w, nrm},
   w = cutoff / sr;
   k = damp;
   g = Tan[π w];
  ];
CalcCoeff2[cutoff_, damp_, sr_] := Block[{w, nrm},
   w = cutoff / sr;
   k = damp;
   s1 = Sin[π w];
   s2 = Sin[2 π w];
   nrm = 1 / (2 + k s2);
   a1 = (2 - 2 s1 * s2) nrm;
   g0 = (s2) nrm;
   g1 = (-2 s1^2 - k s2) nrm;
   g2 = (2 s1^2) nrm;
  ];
Tick1[t_, v0_] := Block[{v1, v2, t1, t2, high, band, low, peak, notch},
   v1 = (ic1eq + g (-ic2eq + v0)) / (1 + g (g + k));
   v2 = (ic2eq + g (ic1eq + ic2eq k + g v0)) / (1 + g (g + k));
   ic1eq = 2 v1 - ic1eq;
   ic2eq = 2 v2 - ic2eq;
   high = v0 - k v1 - v2;
   band = v1;
   low = v2;
   notch = high + low;
   peak = high - low;
   Return[{t, v0, high, band, low, notch, peak}]
  ];
Tick2[t_, v0_] :=
  Block[{v1, v2, t0, tc1, t1, t2, high, band, low, peak, notch},
   t0 = v0 - ic2eq;
   t1 = g0 * t0 + g1 * ic1eq;
   t2 = g2 * t0 + g0 * ic1eq;
   v1 = t1 + ic1eq;
   v2 = t2 + ic2eq;
   ic1eq = ic1eq + 2.0 * t1;
   ic2eq = ic2eq + 2.0 * t2;
```

```
    high = v0 - k v1 - v2;
    band = v1;
    low = v2;
    notch = high + low;
    peak = high - low;
    Return[{t, v0, high, band, low, notch, peak}]
   ];

sr = 44 100.0;
h = 1.0 / sr;
cutoff = 200.0;
damp = 1;

MySaw[x_] := 2 (x - Floor[x] - 0.5);
MyOsc[x_] := 1 MySaw[200 x];

CalcCoeff1[cutoff, damp, sr];
CalcCoeff2[cutoff, damp, sr];
ic1eq = ic2eq = 0;
tp1 = Table[Tick1[t, MyOsc[t]] , {t, 0, 0.01, h}];
ic1eq = ic2eq = 0;
tp2 = Table[Tick2[t, MyOsc[t]] , {t, 0, 0.01, h}];
ListPlot[Table[tp1[[All, {1, i}]], {i, 2, 7}],
 Joined → True, PlotRange → All, PlotLabel → "Tan version"]
ListPlot[Table[tp2[[All, {1, i}]], {i, 2, 7}],
 Joined → True, PlotRange → All, PlotLabel → "Sin version"]
```





This residual shows that the Sin version is slightly more accurate than the Tan version, which will be more important for limited precision implementations

```
tp3 = Table[Flatten[{tp1[[j, 1]], tp1[[j, 2]],
      Table[tp1[[j, i]] - tp2[[j, i]], {i, 3, 7}]}], {j, 1, Length[tp1]}];
ListPlot[Table[tp3[[All, {1, i}]], {i, 3, 7}], Joined → True,
 PlotRange → All, PlotLabel → "residual"]
```



Solve for the mixed output SVF to show relation to regular DF1 coefficients, mixing high, band, low

```
Remove["Global`*"]
hz =
 hs /. Solve[{vhigh == v0 - 2 k v1 - v2, vband == v1, vlow == v2, v1 == ic1eq + g vhigh,
       ic1eq == (2 v1 - ic1eq) z^-1, v2 == ic2eq + g v1, ic2eq == (2 v2 - ic2eq) z^-1,
       hs == (m0 vhigh + m1 vband + m2 vlow) / v0}, {hs},
      {ic1eq, ic2eq, v0, v1, v2, vhigh, vband, vlow}][[1]] // FullSimplify
svfa = Reverse[CoefficientList[Numerator[hz], z]];
svfb = Reverse[CoefficientList[Denominator[hz], z]];
svfa /= svfb[[1]]
svfb /= svfb[[1]]

hzsvf = (svfa[[1]] + svfa[[2]] z^-1 + svfa[[3]] z^-2) / (1 + svfb[[2]] z^-1 + svfb[[3]] z^-2)

hzdf1 = (a0 + a1 z^-1 + a2 z^-2) / (1 - b1 z^-1 - b2 z^-2)

sgkm = FullSimplify[Solve[{a0 == svfa[[1]], a1 == svfa[[2]],
     a2 == svfa[[3]], b1 == svfb[[2]], b2 == svfb[[3]]}, {g, k, m0, m1, m2}]]
assumptions = Element[{w, k, A}, Reals] && w > 0 && w < 1 && k > 0 && k < 2 && A > 0;
SqrSqrtSimplify[eqn_] :=
 FullSimplify[Sqrt[FullSimplify[TrigExpand[(eqn)^2], assumptions]], assumptions]
MatchDF1Coeff[b0_, b1_, b2_, a0_, a1_, a2_] := Block[{w, g, k, m0, m1, m2},
   Table[{g → SqrSqrtSimplify[g /. sgkm[[i]]],
     k → (-1)^(i+1) SqrSqrtSimplify[k /. sgkm[[i]]],
     m0 → SqrSqrtSimplify[m0 /. sgkm[[i]]],
     m1 → (-1)^i SqrSqrtSimplify[m1 /. sgkm[[i]]],
     m2 → SqrSqrtSimplify[m2 /. sgkm[[i]]]}, {i, 1, 2}]
 ]
```

$$\frac{m0\ (-1+z)^2+g\ (1+z)\ (m1\ (-1+z)+g\ m2\ (1+z))}{(-1+z)^2+g^2\ (1+z)^2+2\ g\ k\ \left(-1+z^2\right)}$$

$$\left\{\frac{m0+g\ m1+g^2\ m2}{1+g^2+2\ g\ k},\ \frac{-2\ m0+2\ g^2\ m2}{1+g^2+2\ g\ k},\ \frac{m0-g\ m1+g^2\ m2}{1+g^2+2\ g\ k}\right\}$$

$$\left\{1,\ \frac{-2+2\ g^2}{1+g^2+2\ g\ k},\ \frac{1+g^2-2\ g\ k}{1+g^2+2\ g\ k}\right\}$$

$$\frac{\frac{m0+g\ m1+g^2\ m2}{1+g^2+2\ g\ k}+\frac{m0-g\ m1+g^2\ m2}{(1+g^2+2\ g\ k)\ z^2}+\frac{-2\ m0+2\ g^2\ m2}{(1+g^2+2\ g\ k)\ z}}{1+\frac{1+g^2-2\ g\ k}{(1+g^2+2\ g\ k)\ z^2}+\frac{-2+2\ g^2}{(1+g^2+2\ g\ k)\ z}}$$

$$\frac{a0+\frac{a2}{z^2}+\frac{a1}{z}}{1-\frac{b2}{z^2}-\frac{b1}{z}}$$

$$\left\{\left\{g\to-\frac{\sqrt{-1-b1-b2}}{\sqrt{-1+b1-b2}},\ k\to\frac{1-b2}{\sqrt{-1-b1-b2}\ \sqrt{-1+b1-b2}},\right.\right.$$
$$\left.m0\to\frac{a0-a1+a2}{1-b1+b2},\ m1\to\frac{2\ (a0-a2)}{\sqrt{-1-b1-b2}\ \sqrt{-1+b1-b2}},\ m2\to\frac{a0+a1+a2}{1+b1+b2}\right\},$$
$$\left\{g\to\frac{\sqrt{-1-b1-b2}}{\sqrt{-1+b1-b2}},\ k\to\frac{-1+b2}{\sqrt{-1-b1-b2}\ \sqrt{-1+b1-b2}},\ m0\to\frac{a0-a1+a2}{1-b1+b2},\right.$$
$$\left.\left.m1\to-\frac{2\ (a0-a2)}{\sqrt{-1-b1-b2}\ \sqrt{-1+b1-b2}},\ m2\to\frac{a0+a1+a2}{1+b1+b2}\right\}\right\}$$

Calculate g, k, m0, m1, m2 for some specific DF1 coefficients.

This is the completely wrong way around to think of this stuff since you should be using the continuous analog filter prototypes since the equations are soooo much easier, as is seen by the results of the coefficients, they are fairly elegant and make sense, where the DF1 coeffs are completely abastract

DF1 low pass

```
cw = Cos[π w];
sw = Sin[π w];
alpha = k * sw / 2;

b0 = 1 + alpha;
b1 = (-2 * cw) / b0;
b2 = (1 - alpha) / b0;
a0 = 1 / 2 (1 - cw) / b0;
a1 = (1 - cw) / b0;
a2 = 1 / 2 (1 - cw) / b0;

MatchDF1Coeff[b0, b1, b2, a0, a1, a2]
```

$$\left\{\left\{g\to\text{Tan}\left[\frac{\pi\ w}{2}\right],\ k\to\frac{k}{2},\ m0\to0,\ m1\to0,\ m2\to1\right\},\right.$$
$$\left.\left\{g\to\text{Tan}\left[\frac{\pi\ w}{2}\right],\ k\to-\frac{k}{2},\ m0\to0,\ m1\to0,\ m2\to1\right\}\right\}$$

DF1 high pass

```
cw = Cos[π w];
sw = Sin[π w];
alpha = k * sw / 2;

b0 = 1 + alpha;
b1 = (-2 * cw) / b0;
b2 = (1 - alpha) / b0;
a0 = 1 / 2 (1 + cw) / b0;
a1 = - (1 + cw) / b0;
a2 = 1 / 2 (1 + cw) / b0;

MatchDF1Coeff[b0, b1, b2, a0, a1, a2]
```

$$\left\{\left\{g \to \text{Tan}\left[\frac{\pi w}{2}\right], k \to \frac{k}{2}, m0 \to 1, m1 \to 0, m2 \to 0\right\}, \left\{g \to \text{Tan}\left[\frac{\pi w}{2}\right], k \to -\frac{k}{2}, m0 \to 1, m1 \to 0, m2 \to 0\right\}\right\}$$

DF1 bell

```
cw = Cos[π w];
sw = Sin[π w];
alpha = k * sw / 2;

b0 = 1 + alpha / A;
b1 = (-2 * cw) / b0;
b2 = (1 - alpha / A) / b0;
a0 = (1 + alpha * A) / b0;
a1 = (-2 * cw) / b0;
a2 = (1 - alpha * A) / b0;

MatchDF1Coeff[b0, b1, b2, a0, a1, a2]
```

$$\left\{\left\{g \to \text{Tan}\left[\frac{\pi w}{2}\right], k \to \frac{k}{2 A}, m0 \to 1, m1 \to -A k, m2 \to 1\right\}, \left\{g \to \text{Tan}\left[\frac{\pi w}{2}\right], k \to -\frac{k}{2 A}, m0 \to 1, m1 \to A k, m2 \to 1\right\}\right\}$$

DF1 low shelf

```
cw = Cos[π w];
sw = Sin[π w];
alpha = k * sw / 2;

sqrtA = Sqrt[A];
b0 = ((A + 1) + (A - 1) * cw + 2 * sqrtA * alpha);
b1 = (-2 * ((A - 1) + (A + 1) * cw)) / b0;
b2 = ((A + 1) + (A - 1) * cw - 2 * sqrtA * alpha) / b0;
a0 = (A * ((A + 1) - (A - 1) * cw + 2 * sqrtA * alpha)) / b0;
a1 = (2 * A * ((A - 1) - (A + 1) * cw)) / b0;
a2 = (A * ((A + 1) - (A - 1) * cw - 2 * sqrtA * alpha)) / b0;

MatchDF1Coeff[b0, b1, b2, a0, a1, a2]
```

$$\left\{\left\{g \to \frac{\text{Tan}\left[\frac{\pi w}{2}\right]}{\sqrt{A}},\ k \to \frac{k}{2},\ m0 \to 1,\ m1 \to -A\,k,\ m2 \to A^2\right\},\ \left\{g \to \frac{\text{Tan}\left[\frac{\pi w}{2}\right]}{\sqrt{A}},\ k \to -\frac{k}{2},\ m0 \to 1,\ m1 \to A\,k,\ m2 \to A^2\right\}\right\}$$

DF1 high shelf

```
cw = Cos[π w];
sw = Sin[π w];
alpha = k * sw / 2;

sqrtA = Sqrt[A];
b0 = (A + 1) - (A - 1) * cw + 2 * sqrtA * alpha;
b1 = 2 * ((A - 1) - (A + 1) * cw) / b0;
b2 = ((A + 1) - (A - 1) * cw - 2 * sqrtA * alpha) / b0;
a0 = A * ((A + 1) + (A - 1) * cw + 2 * sqrtA * alpha) / b0;
a1 = -2 * A * ((A - 1) + (A + 1) * cw) / b0;
a2 = A * ((A + 1) + (A - 1) * cw - 2 * sqrtA * alpha) / b0;

MatchDF1Coeff[b0, b1, b2, a0, a1, a2]
```

$$\left\{\left\{g \to \sqrt{A}\ \text{Tan}\left[\frac{\pi w}{2}\right],\ k \to \frac{k}{2},\ m0 \to A^2,\ m1 \to -A\,k,\ m2 \to 1\right\},\ \left\{g \to \sqrt{A}\ \text{Tan}\left[\frac{\pi w}{2}\right],\ k \to -\frac{k}{2},\ m0 \to A^2,\ m1 \to A\,k,\ m2 \to 1\right\}\right\}$$